# $Id$
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=xorg-xedit
pkgver=1.2.2
pkgrel=1
pkgdesc="Simple text editor for X"
arch=('i686' 'x86_64')
url="https://xorg.freedesktop.org/"
license=('custom')
depends=('libxaw' 'libxmu' 'libxt' 'libx11')
makedepends=('xorg-util-macros')
source=(https://xorg.freedesktop.org/archive/individual/app/xedit-${pkgver}.tar.bz2{,.sig})
sha256sums=('69aa42885dfc06332ca22eb01cc7187e49206e6d65b74113a8ee4cc345fc2927'
            'SKIP')
validpgpkeys=('3BB639E56F861FA2E86505690FDD682D974CA72A') # "Matt Turner <mattst88@gmail.com>"

build() {
  cd "${srcdir}/xedit-${pkgver}"
  ./configure --prefix=/usr
  make
}

package() {
  cd "${srcdir}/xedit-${pkgver}"
  make DESTDIR="${pkgdir}" install
  install -Dm644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/COPYING"
}
